

NAG Fortran Library Routine Document

F08NBF (DGEEVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08NBF (DGEEVX) computes the eigenvalues and, optionally, the left and/or right eigenvectors for an n by n real non-symmetric matrix A .

Optionally, it also computes a balancing transformation to improve the conditioning of the eigenvalues and eigenvectors, reciprocal condition numbers for the eigenvalues, and reciprocal condition numbers for the right eigenvectors.

2 Specification

```

SUBROUTINE F08NBF (BALANC, JOBVL, JOBVR, SENSE, N, A, LDA, WR, WI, VL,
1             LDVL, VR, LDVR, ILO, IHI, SCALE, ABNRM, RCONDE,
2             RCONDV, WORK, LWORK, IWORK, INFO)
    INTEGER          N, LDA, LDVL, LDVR, ILO, IHI, LWORK, IWORK(*), INFO
    double precision A(LDA,*), WR(*), WI(*), VL(LDVL,*), VR(LDVR,*),
1             SCALE(*), ABNRM, RCONDE(*), RCONDV(*), WORK(*)
    CHARACTER*1     BALANC, JOBVL, JOBVR, SENSE

```

The routine may be called by its LAPACK name *dgeevx*.

3 Description

The right eigenvector v_j of A satisfies

$$Av_j = \lambda_j v_j$$

where λ_j is the j th eigenvalue of A . The left eigenvector u_j of A satisfies

$$u_j^H A = \lambda_j u_j^H$$

where u_j^H denotes the conjugate transpose of u_j .

Balancing a matrix A comprises two operations. The first is to permute the rows and columns of A to make it 'more nearly' upper triangular (closer to Schur form): $A' = PAP^T$, where P is a permutation matrix. The second operation is applying a diagonal similarity transformation DAD^{-1} , where D is a diagonal matrix, with the aim of making its rows and columns closer in norm and the condition numbers of its eigenvalues and eigenvectors smaller. The computed reciprocal condition numbers correspond to the balanced matrix. Permuting rows and columns will not change the condition numbers (in exact arithmetic) but diagonal scaling will. For further explanation of balancing, see Section 4.8.1.2 of Anderson *et al.* (1999).

Following the optional balancing, the matrix A is first reduced to upper Hessenberg form by means of orthogonal similarity transformations, and the QR algorithm is then used to further reduce the matrix to quasi-upper triangular Schur form, T , with 1 by 1 and 2 by 2 blocks on the main diagonal. The eigenvalues are computed from T , the 2 by 2 blocks corresponding to complex conjugate pairs and, optionally, the eigenvectors of T are computed and backtransformed to the eigenvectors of A .

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: BALANC – CHARACTER*1 *Input*
- On entry:* indicates how the input matrix should be diagonally scaled and/or permuted to improve the conditioning of its eigenvalues.
- BALANC = 'N'
- Do not diagonally scale or permute.
- BALANC = 'P'
- Perform permutations to make the matrix more nearly upper triangular. Do not diagonally scale.
- BALANC = 'S'
- Diagonally scale the matrix, i.e., replace A by DAD^{-1} , where D is a diagonal matrix chosen to make the rows and columns of A more equal in norm. Do not permute.
- BALANC = 'B'
- Both diagonally scale and permute A .
- Computed reciprocal condition numbers will be for the matrix after balancing and/or permuting. Permuting does not change condition numbers (in exact arithmetic), but balancing does.
- 2: JOBVL – CHARACTER*1 *Input*
- On entry:* if JOBVL = 'N', the left eigenvectors of A are not computed.
- If JOBVL = 'V', the left eigenvectors of A are computed.
- If SENSE = 'E' or 'B', JOBVL must be set to JOBVL = 'V'.
- 3: JOBVR – CHARACTER*1 *Input*
- On entry:* if JOBVR = 'N', the right eigenvectors of A are not computed.
- If JOBVR = 'V', the right eigenvectors of A are computed.
- If SENSE = 'E' or 'B', JOBVR must be set to JOBVR = 'V'.
- 4: SENSE – CHARACTER*1 *Input*
- On entry:* determines which reciprocal condition numbers are computed.
- SENSE = 'N'
- None are computed.
- SENSE = 'E'
- Computed for eigenvalues only.
- SENSE = 'V'
- Computed for right eigenvectors only.
- SENSE = 'B'
- Computed for eigenvalues and right eigenvectors.
- If SENSE = 'E' or 'B', both left and right eigenvectors must also be computed (JOBVL = 'V' and JOBVR = 'V').

- 5: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 6: A(LDA,*) – **double precision** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n matrix A .
On exit: has been overwritten. If $\text{JOBVL} = 'V'$ or $\text{JOBVR} = 'V'$, A contains the real Schur form of the balanced version of the input matrix A .
- 7: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08NBF (DGEEVX) is called.
Constraint: $\text{LDA} \geq \max(1, N)$.
- 8: WR(*) – **double precision** array *Output*
Note: the dimension of the array WR must be at least $\max(1, N)$.
On exit: see the description of WI .
- 9: WI(*) – **double precision** array *Output*
Note: the dimension of the array WR and WI must be at least $\max(1, N)$.
On exit: WR and WI contain the real and imaginary parts, respectively, of the computed eigenvalues. Complex conjugate pairs of eigenvalues appear consecutively with the eigenvalue having the positive imaginary part first.
- 10: VL(LDVL,*) – **double precision** array *Output*
Note: the second dimension of the array VL must be at least $\max(1, N)$.
On exit: if $\text{JOBVL} = 'V'$, the left eigenvectors u_j are stored one after another in the columns of VL , in the same order as their corresponding eigenvalues.
 If $\text{JOBVL} = 'N'$, VL is not referenced.
 If the j th eigenvalue is real, then $u_j = \text{VL}(:,j)$, the j th column of VL .
 If the j th and $(j+1)$ st eigenvalues form a complex conjugate pair, then $u_j = \text{VL}(:,j) + i \times \text{VL}(:,j+1)$ and $u_{j+1} = \text{VL}(:,j) - i \times \text{VL}(:,j+1)$.
- 11: LDVL – INTEGER *Input*
On entry: the first dimension of the array VL as declared in the (sub)program from which F08NBF (DGEEVX) is called.
Constraints:
 if $\text{JOBVL} = 'V'$, $\text{LDVL} \geq \max(1, N)$;
 $\text{LDVL} \geq 1$ otherwise.
- 12: VR(LDVR,*) – **double precision** array *Output*
Note: the second dimension of the array VR must be at least $\max(1, N)$.
On exit: if $\text{JOBVR} = 'V'$, the right eigenvectors v_j are stored one after another in the columns of VR , in the same order as their corresponding eigenvalues.
 If $\text{JOBVR} = 'N'$, VR is not referenced.
 If the j th eigenvalue is real, then $v_j = \text{VR}(:,j)$, the j th column of VR .

If the j th and $(j+1)$ st eigenvalues form a complex conjugate pair, then $v_j = \text{VR}(:,j) + i \times \text{VR}(:,j+1)$ and $v_{j+1} = \text{VR}(:,j) - i \times \text{VR}(:,j+1)$.

13: LDVR – INTEGER *Input*

On entry: the first dimension of the array VR as declared in the (sub)program from which F08NBF (DGEEVX) is called.

Constraints:

if JOBVR = 'V', LDVR \geq max(1, N);
LDVR \geq 1 otherwise.

14: ILO – INTEGER *Output*

15: IHI – INTEGER *Output*

On exit: ILO and IHI are integer values determined when A was balanced. The balanced A has $a_{ij} = 0$ if $i > j$ and $j = 1, \dots, \text{ILO} - 1$ or $i = \text{IHI} + 1, \dots, N$.

16: SCALE(*) – *double precision* array *Output*

Note: the dimension of the array SCALE must be at least max(1, N).

On exit: details of the permutations and scaling factors applied when balancing A .

If p_j is the index of the row and column interchanged with row and column j , and d_j is the scaling factor applied to row and column j , then

SCALE(j) = p_j , for $j = 1, \dots, \text{ILO} - 1$;

SCALE(j) = d_j , for $j = \text{ILO}, \dots, \text{IHI}$;

SCALE(j) = p_j , for $j = \text{IHI} + 1, \dots, N$.

The order in which the interchanges are made is N to IHI + 1, then 1 to ILO - 1.

17: ABNRM – *double precision* *Output*

On exit: the 1-norm of the balanced matrix (the maximum of the sum of absolute values of elements of any column).

18: RCONDE(*) – *double precision* array *Output*

Note: the dimension of the array RCONDE must be at least max(1, N).

On exit: RCONDE(j) is the reciprocal condition number of the j th eigenvalue.

19: RCONDV(*) – *double precision* array *Output*

Note: the dimension of the array RCONDV must be at least max(1, N).

On exit: RCONDV(j) is the reciprocal condition number of the j th right eigenvector.

20: WORK(*) – *double precision* array *Workspace*

Note: the dimension of the array WORK must be at least max(1, LWORK).

On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimal performance.

21: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08NBF (DGEEVX) is called.

If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK must generally be larger than the minimum, increase LWORK by, say, $N \times nb$, where nb is the optimal **block size** for F08NEF (DGEHRD).

Constraints:

- if SENSE = 'N' or 'E', LWORK $\geq \max(1, 2 \times N)$;
- if JOBVL = 'V' or JOBVR = 'V', LWORK $\geq \max(1, 3 \times N)$;
- if SENSE = 'V' or 'B', LWORK $\geq \max(1, N \times (N + 6))$.

22: IWORK(*) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, 2 \times N - 1)$.

If SENSE = 'N' or 'E', IWORK is not referenced.

23: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , the QR algorithm failed to compute all the eigenvalues, and no eigenvectors or condition numbers have been computed; elements 1 : ILO – 1 and $i + 1$: N of WR and WI contain eigenvalues which have converged.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*. See Section 4.8 of Anderson *et al.* (1999) for further details.

8 Further Comments

Each eigenvector is normalized to have Euclidean norm equal to unity and the element of largest absolute value real and positive.

The total number of floating-point operations is proportional to n^3 .

The complex analogue of this routine is F08NPF (ZGEEVX).

9 Example

This example finds all the eigenvalues and right eigenvectors of the matrix

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix},$$

together with estimates of the condition number and forward error bounds for each eigenvalue and eigenvector. The option to balance the matrix is used. In order to compute the condition numbers of the eigenvalues, the left eigenvectors also have to be computed, but they are not printed out in this example.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

```

*      F08NBF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          NB, NMAX
PARAMETER        (NB=64,NMAX=10)
INTEGER          LDA, LDVL, LDVR, LWORK
PARAMETER        (LDA=NMAX,LDVL=NMAX,LDVR=NMAX,LWORK=(2+NB)*NMAX)
*      .. Local Scalars ..
COMPLEX *16      EIG
DOUBLE PRECISION ABNRM, EPS, ERBND, RCND, TOL
INTEGER          I, IHI, ILO, INFO, J, LWKOPT, N
*      .. Local Arrays ..
DOUBLE PRECISION A(LDA,NMAX), RCONDE(NMAX), RCONDV(NMAX),
+                SCALE(NMAX), VL(LDVL,NMAX), VR(LDVR,NMAX),
+                WI(NMAX), WORK(LWORK), WR(NMAX)
INTEGER          IWORK(2*NMAX-2)
*      .. External Functions ..
DOUBLE PRECISION X02AJF
EXTERNAL         X02AJF
*      .. External Subroutines ..
EXTERNAL         DGEEVX
*      .. Intrinsic Functions ..
INTRINSIC        CMPLX
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08NBF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read the matrix A from data file
*
READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*      Solve the eigenvalue problem
*
CALL DGEEVX('Balance','Vectors (left)','Vectors (right)',
+          'Both reciprocal condition numbers',N,A,LDA,WR,WI,
+          VL,LDVL,VR,LDVR,ILO,IHI,SCALE,ABNRM,RCONDE,RCONDV,
+          WORK,LWORK,IWORK,INFO)
*
IF (INFO.EQ.0) THEN
*
*      Compute the machine precision
*
EPS = X02AJF()
TOL = EPS*ABNRM
*
*      Print the eigenvalues and vectors, and associated condition
*      number and bounds
*
DO 20 J = 1, N
*
*      Print information on jth eigenvalue
*
WRITE (NOUT,*)
IF (WI(J).EQ.0.0D0) THEN
WRITE (NOUT,99999) 'Eigenvalue(', J, ') = ', WR(J)
ELSE
EIG = CMPLX(WR(J),WI(J),KIND=KIND(EPS))
WRITE (NOUT,99998) 'Eigenvalue(', J, ') = ', EIG
END IF
RCND = RCONDE(J)

```

```

        WRITE (NOUT,*)
        WRITE (NOUT,99997) 'Reciprocal condition number = ', RCND
        IF (RCND.GT.0.0D0) THEN
            ERBND = TOL/RCND
            WRITE (NOUT,99997) 'Error bound = ',
+             ERBND
        ELSE
            WRITE (NOUT,*) 'Error bound is infinite'
        END IF
*
*       Print information on jth eigenvector
*
        WRITE (NOUT,*)
        WRITE (NOUT,99996) 'Eigenvector(', J, ' )'
        IF (WI(J).EQ.0.0D0) THEN
            WRITE (NOUT,99995) (VR(I,J),I=1,N)
        ELSE IF (WI(J).GT.0.0D0) THEN
            WRITE (NOUT,99994) (VR(I,J),VR(I,J+1),I=1,N)
        ELSE
            WRITE (NOUT,99994) (VR(I,J-1),-VR(I,J),I=1,N)
        END IF
        RCND = RCONDV(J)
        WRITE (NOUT,*)
        WRITE (NOUT,99997) 'Reciprocal condition number = ', RCND
        IF (RCND.GT.0.0D0) THEN
            ERBND = TOL/RCND
            WRITE (NOUT,99997) 'Error bound = ',
+             ERBND
        ELSE
            WRITE (NOUT,*) 'Error bound is infinite'
        END IF
20    CONTINUE
    ELSE
        WRITE (NOUT,*)
        WRITE (NOUT,99993) 'Failure in DGEEVX. INFO = ', INFO
    END IF
*
*       Print workspace information
*
        LWKOPT = WORK(1)
        IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99992) 'Optimum workspace required = ', LWKOPT,
+             'Workspace provided = ', LWORK
        END IF
    ELSE
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'NMAX too small'
    END IF
    STOP
*
99999 FORMAT (1X,A,I2,A,1P,E11.4)
99998 FORMAT (1X,A,I2,A,'(',1P,E11.4,',',1P,E11.4,')')
99997 FORMAT (1X,A,1P,E8.1)
99996 FORMAT (1X,A,I2,A)
99995 FORMAT (1X,1P,E11.4)
99994 FORMAT (1X,'(',1P,E11.4,',',1P,E11.4,')')
99993 FORMAT (1X,A,I4)
99992 FORMAT (1X,A,I5,/1X,A,I5)
    END

```

9.2 Program Data

F08NBF Example Program Data

```

4           :Value of N
0.35      0.45      -0.14      -0.17
0.09      0.07      -0.54      0.35
-0.44     -0.33     -0.03      0.17
0.25     -0.32     -0.13      0.11 :End of matrix A

```

9.3 Program Results

F08NBF Example Program Results

Eigenvalue(1) = 7.9948E-01

Reciprocal condition number = 9.9E-01
Error bound = 1.3E-16

Eigenvector(1)
-6.5509E-01
-5.2363E-01
5.3622E-01
-9.5607E-02

Reciprocal condition number = 6.3E-01
Error bound = 2.1E-16

Eigenvalue(2) = (-9.9412E-02, 4.0079E-01)

Reciprocal condition number = 7.0E-01
Error bound = 1.8E-16

Eigenvector(2)
(-1.9330E-01, 2.5463E-01)
(2.5186E-01, -5.2240E-01)
(9.7182E-02, -3.0838E-01)
(6.7595E-01, 0.0000E+00)

Reciprocal condition number = 4.0E-01
Error bound = 3.3E-16

Eigenvalue(3) = (-9.9412E-02, -4.0079E-01)

Reciprocal condition number = 7.0E-01
Error bound = 1.8E-16

Eigenvector(3)
(-1.9330E-01, -2.5463E-01)
(2.5186E-01, 5.2240E-01)
(9.7182E-02, 3.0838E-01)
(6.7595E-01, 0.0000E+00)

Reciprocal condition number = 4.0E-01
Error bound = 3.3E-16

Eigenvalue(4) = -1.0066E-01

Reciprocal condition number = 5.7E-01
Error bound = 2.3E-16

Eigenvector(4)
1.2533E-01
3.3202E-01
5.9384E-01
7.2209E-01

Reciprocal condition number = 3.1E-01
Error bound = 4.2E-16
